Algorithmen des wissenschaftlichen Rechnens II
Übungsblatt 5
Quadratur mit der Kombinationstechnik

```
> restart;
> with(plots):
> with(plottools):
```
```
Warning, the name changecoords has been redefined

Warning, the assigned name arrow now has a global binding
```

Ein paar generelle Optionen zum Plotten:

```
> poptions := thickness=2,
              symbol=DIAMOND, symbolsize=20:
```

Eine Quadraturformel wird hier dargestellt als ein assoziatives Array (maple: table) mit den Koordinaten der Auswertestellen (2-komponentige Listen) als Schlüssel und den Gewichten als Werten.
Beim Tabellenerzeugen geben wir 'sparse' als Indexfunktion an - das bewirkt, dass wir das Gewicht 0 für alle Punkte bekommen, die nicht explitit einen anderen Wert bekommen haben (für's Kombinieren wird das praktisch sein...).

```
>
```

Trapezregel

So erzeugt man z.B. die Trapezregel mit N1 Intervallen in x1-Richtung und N2 Intervallen in x2-Richtung

```
> trapezregel := proc(N1,N2)
     local tr,i,j,s;
     tr := table(sparse);
     s := 1/(N1*N2);
     for i from 0 to N1 do
        for j from 0 to N2 do
           tr[i/N1,j/N2] := s;
           if i=0 or i=N1 then tr[i/N1,j/N2] := tr[i/N1,j/N2]/2
  end if;
           if j=0 or j=N2 then tr[i/N1,j/N2] := tr[i/N1,j/N2]/2
  end if
        end do
     end do;
     return tr;
  end proc:
```

```
> t := trapezregel(3,4);
```

$$t := tr$$

Das ist in der Tabelle drin:

```
> eval(t);
```

$$\text{table}(sparse, [(1,1) = \frac{1}{48}, \left(\frac{1}{3}, \frac{1}{2}\right) = \frac{1}{12}, \left(1, \frac{1}{2}\right) = \frac{1}{24}, \left(\frac{2}{3}, \frac{1}{4}\right) = \frac{1}{12}, (0,0) = \frac{1}{48}, \left(1, \frac{3}{4}\right) = \frac{1}{24},$$

$$\left(0, \frac{1}{4}\right) = \frac{1}{24}, \left(\frac{1}{3}, 1\right) = \frac{1}{24}, \left(\frac{2}{3}, 1\right) = \frac{1}{24}, \left(\frac{2}{3}, \frac{1}{2}\right) = \frac{1}{12}, (1, 0) = \frac{1}{48}, (0, 1) = \frac{1}{48}, \left(\frac{1}{3}, 0\right) = \frac{1}{24},$$
$$\left(\frac{1}{3}, \frac{1}{4}\right) = \frac{1}{12}, \left(\frac{2}{3}, 0\right) = \frac{1}{24}, \left(1, \frac{1}{4}\right) = \frac{1}{24}, \left(0, \frac{1}{2}\right) = \frac{1}{24}, \left(\frac{1}{3}, \frac{3}{4}\right) = \frac{1}{12}, \left(0, \frac{3}{4}\right) = \frac{1}{24},$$
$$\left(\frac{2}{3}, \frac{3}{4}\right) = \frac{1}{12}$$

])

```
> indices(t);
```

$$[1, 1], \left[\frac{1}{3}, \frac{1}{2}\right], \left[1, \frac{1}{2}\right], \left[\frac{2}{3}, \frac{1}{4}\right], [0, 0], \left[1, \frac{3}{4}\right], \left[0, \frac{1}{4}\right], \left[\frac{1}{3}, 1\right], \left[\frac{2}{3}, 1\right], \left[\frac{2}{3}, \frac{1}{2}\right], [1, 0], [0, 1],$$
$$\left[\frac{1}{3}, 0\right], \left[\frac{1}{3}, \frac{1}{4}\right], \left[\frac{2}{3}, 0\right], \left[1, \frac{1}{4}\right], \left[0, \frac{1}{2}\right], \left[\frac{1}{3}, \frac{3}{4}\right], \left[0, \frac{3}{4}\right], \left[\frac{2}{3}, \frac{3}{4}\right]$$

Und das ist die tyische Schleife über alle Tabelleneinträge:

```
> for i in indices(t) do
      printf("Schlüssel: %a, Wert: %a\n", i, t[op(i)])
  end do;
Schlüssel: [1, 1], Wert: 1/48
Schlüssel: [1/3, 1/2], Wert: 1/12
Schlüssel: [1, 1/2], Wert: 1/24
Schlüssel: [2/3, 1/4], Wert: 1/12
Schlüssel: [0, 0], Wert: 1/48
Schlüssel: [1, 3/4], Wert: 1/24
Schlüssel: [0, 1/4], Wert: 1/24
Schlüssel: [1/3, 1], Wert: 1/24
Schlüssel: [2/3, 1], Wert: 1/24
Schlüssel: [2/3, 1/2], Wert: 1/12
Schlüssel: [1, 0], Wert: 1/48
Schlüssel: [0, 1], Wert: 1/48
Schlüssel: [1/3, 0], Wert: 1/24
Schlüssel: [1/3, 1/4], Wert: 1/12
Schlüssel: [2/3, 0], Wert: 1/24
Schlüssel: [1, 1/4], Wert: 1/24
Schlüssel: [0, 1/2], Wert: 1/24
Schlüssel: [1/3, 3/4], Wert: 1/12
Schlüssel: [0, 3/4], Wert: 1/24
Schlüssel: [2/3, 3/4], Wert: 1/12
```

[0,0] kommt i dieser Trapezregel vor, [1/100,1/100] aber nicht:

```
> t[0,0], t[1/100,1/100];
```

$$\frac{1}{48}, 0$$

```
>
```

Bild malen

Diese Prozedur malt die Gewichte der Quadraturregel qr an den richtigen Stellen hin:

```
> bild := proc(qr)
    local i;
    display([
        rectangle([0,0],[1,1],poptions),
        pointplot([seq(i, i=[indices(qr)])],poptions),
```

```
        seq(textplot([op(i),sprintf("  %a",qr[op(i)])],
                     align=RIGHT,poptions),
            i=[indices(qr)])
        ], font=[HELVETICA,12], axes=NONE, scaling=CONSTRAINED)
  end proc:
> bild(trapezregel(5,7));
```

1/140	1/70	1/70	1/70	1/70	1/140
1/70	1/35	1/35	1/35	1/35	1/70
1/70	1/35	1/35	1/35	1/35	1/70
1/70	1/35	1/35	1/35	1/35	1/70
1/70	1/35	1/35	1/35	1/35	1/70
1/70	1/35	1/35	1/35	1/35	1/70
1/70	1/35	1/35	1/35	1/35	1/70
1/140	1/70	1/70	1/70	1/70	1/140

```
>
```

Quadraturregel anwenden

Vielleicht wollen wir unsere Quadraturregel auch mal auf eine Funktion anwenden - das ist einfach:

```
> anwenden := (qr, f) -> add(qr[op(i)]*f(op(i)),
  i=[indices(qr)]):

> anwenden(trapezregel(2,3), f);
```

$$\frac{1}{24}f(1,1)+\frac{1}{12}f\left(0,\frac{2}{3}\right)+\frac{1}{12}f\left(1,\frac{2}{3}\right)+\frac{1}{24}f(0,0)+\frac{1}{12}f\left(\frac{1}{2},0\right)+\frac{1}{6}f\left(\frac{1}{2},\frac{2}{3}\right)+\frac{1}{24}f(1,0)$$
$$+\frac{1}{24}f(0,1)+\frac{1}{12}f\left(\frac{1}{2},1\right)+\frac{1}{12}f\left(0,\frac{1}{3}\right)+\frac{1}{12}f\left(1,\frac{1}{3}\right)+\frac{1}{6}f\left(\frac{1}{2},\frac{1}{3}\right)$$

```
> anwenden(trapezregel(2,3), 1);
```

$$1$$

```
> evalf(anwenden(trapezregel(10,20), (x,y) ->
  16*x*(1-x)*y*(1-y)));
```

$$0.4389000000$$

```
> evalf(int(int(16*x*(1-x)*y*(1-y), x=0..1),y=0..1));
```

$$0.4444444444$$

```
>
```

Kombinationstechnik

Neue Quadraturregel durch Linearkombination von zwei Quadraturregeln qr1 und qr2: qr(f) := alpha1*qr1(f) + alpha2*qr2(f):

```
> kombinieren := proc(qr1, alpha1, qr2, alpha2)
    local qr, i;
    qr := table(sparse);
    for i in {indices(qr1)} union {indices(qr2)} do
       qr[op(i)] := alpha1*qr1[op(i)] + alpha2*qr2[op(i)]
    end do;
    return qr;
  end proc:
> bild(kombinieren(trapezregel(4,2), 1, trapezregel(2,4), -1));
```



Damit können wir die Summation für die Kombinationstechnik durchführen:

```
> kombiregel := proc(n)
    local qr, i;
    qr := table(sparse);
    for i to n do
      qr := kombinieren(qr, 1, trapezregel(2^i, 2^(n+1-i)),
  1);
    end do;
    for i to n-1 do
      qr := kombinieren(qr, 1, trapezregel(2^i, 2^(n-i)), -1);
    end do;
    return qr
  end proc:
> bild(kombiregel(2));
```




```
> bild(kombiregel(3));
```




```
>
```

Ergänzung: Gewichte bei der Dünngitter-Quadratur

Das ist die Dünngitter-Quadraturregel, für die die Diskrepanz relevant ist (alle Punkte haben das gleiche Gewicht):

```
> duenngitter_mc := proc(n)
     local qr, l1, l2, N1, N2, i, j, npkt;
     qr := table(sparse);
     for l1 to n do
        N1 := 2^l1;
        for l2 to n+1-l1 do
           N2 := 2^l2;
           for i to N1-1 by 2 do
              for j to N2-1 by 2 do
                 qr[i/N1,j/N2] := 1;
              end do
```

```
            end do
         end do
      end do;
      npkt := nops([indices(qr)]);
      for i in [indices(qr)] do
         qr[op(i)] := 1/npkt;
      end do;
      return qr
   end proc:
> bild(duenngitter_mc(3));
```



Diese Quadraturregel lassen wir nun gegen die Kombinationstechnik (bzw. einen Archimedes, der gerade an den Dünngitterpunkten auswertet: das ist das gleiche Ergebnis wie bei der Kombinationstechnik) antreten.
Dazu nehmen wir die bewährte Beispielfunktion:

```
> u := (x,y) -> 16*x*(1-x)*y*(1-y);
> ref := int(int(u(x,y),x=0..1),y=0..1);
```

$$u := (x, y) \to 16\, x\, (1 - x)\, y\, (1 - y)$$

$$ref := \frac{4}{9}$$

Das sind für Level 2..10 die Fehler bei den beiden Quadraturregeln (im :Bild: Zweierlogarithmus des Fehlers gegen Level aufgetragen)

```
> fehler_dg :=
  [seq(anwenden(duenngitter_mc(i),u)-ref,i=2..10)];
> fehler_kombi := [seq(anwenden(kombiregel(i),u)-ref,i=2..10)];
> pd := pointplot([seq([i,log[2](abs(fehler_dg[i-1]))],
  i=2..10)],
                  poptions, connect=true, color = RED):
> pk := pointplot([seq([i,log[2](abs(fehler_kombi[i-1]))],
  i=2..10)],
                  poptions, connect=true, color = BLUE):
> display([pd,pk]);
```

$$fehler\_dg := \left[\frac{16}{45}, \frac{151}{612}, \frac{323}{1764}, \frac{887}{6192}, \frac{899}{7704}, \frac{43427}{442944}, \frac{87121}{1032768}, \frac{698017}{9439488}, \frac{87317}{1327248}\right]$$

$$fehler\_kombi := \left[\frac{-5}{72}, \frac{-13}{576}, \frac{-1}{144}, \frac{-19}{9216}, \frac{-11}{18432}, \frac{-25}{147456}, \frac{-7}{147456}, \frac{-31}{2359296}, \frac{-17}{4718592}\right]$$


-2
-4
-6
-8
-10
-12
-14
-16
-18
2
4
6
8
10
V
V